

Network Working Group
Internet-Draft
Intended status: Standards Track
Expires: October 9, 2011

C. Daboo
Apple Inc.
April 7, 2011

**New Properties for iCalendar
draft-daboo-icalendar-extensions-04**

Abstract

This document defines a set of new properties for iCalendar data.

Status of this Memo

This Internet-Draft is submitted in full conformance with the provisions of [BCP 78](#) and [BCP 79](#).

Internet-Drafts are working documents of the Internet Engineering Task Force (IETF). Note that other groups may also distribute working documents as Internet-Drafts. The list of current Internet-Drafts is at <http://datatracker.ietf.org/drafts/current/>.

Internet-Drafts are draft documents valid for a maximum of six months and may be updated, replaced, or obsoleted by other documents at any time. It is inappropriate to use Internet-Drafts as reference material or to cite them other than as "work in progress."

This Internet-Draft will expire on October 9, 2011.

Copyright Notice

Copyright (c) 2011 IETF Trust and the persons identified as the document authors. All rights reserved.

This document is subject to [BCP 78](#) and the IETF Trust's Legal Provisions Relating to IETF Documents (<http://trustee.ietf.org/license-info>) in effect on the date of publication of this document. Please review these documents carefully, as they describe your rights and restrictions with respect to this document. Code Components extracted from this document must include Simplified BSD License text as described in Section 4.e of the Trust Legal Provisions and are provided without warranty as described in the Simplified BSD License.

Table of Contents

1.	Introduction	3
2.	Conventions Used in This Document	3
3.	Modifications to Calendar Components	3
4.	Properties	4
4.1.	NAME Property	5
4.2.	DESCRIPTION Property	6
4.3.	UID Property	6
4.4.	URL Property	6
4.5.	TIMEZONE-ID Property	7
4.6.	REFRESH-INTERVAL Property	7
4.7.	COLOR Property	8
4.8.	IMAGE Property	9
4.9.	BROADCAST Property	10
4.10.	CONFERENCE Property	12
5.	Property Parameters	14
5.1.	ALTURI Property Parameter	14
5.2.	DISPLAY Property Parameter	14
5.3.	FEATURE Property Parameter	15
5.4.	INFO Property Parameter	16
5.5.	REGION Property Parameter	16
6.	Security Considerations	17
7.	IANA Considerations	17
7.1.	Property Registrations	17
7.2.	Parameter Registrations	18
7.3.	Display Types Registry	18
7.4.	Feature Types Registry	18
8.	Acknowledgments	19
9.	References	19
9.1.	Normative References	19
9.2.	Informative References	19
Appendix A.	Change History (To be removed by RFC Editor before publication)	19
	Author's Address	20

Daboo

Expires October 9, 2011

[Page 2]

1. Introduction

The iCalendar [[RFC5545](#)] data format is used to represent calendar data and is used with iTIP [[RFC5546](#)] to handle scheduling operations between calendar users. iCalendar is in widespread use, and in accordance with provisions in that specification, extension elements have been added by various vendors to the data format in order to support and enhance capabilities. This specification collects a number of these ad-hoc extensions and uses the new IANA registry capability defined in [[RFC5545](#)] to register standard variants with clearly defined definitions and semantics. In addition, some new elements are introduced for features that vendors have recently been requesting.

2. Conventions Used in This Document

The key words "MUST", "MUST NOT", "REQUIRED", "SHALL", "SHALL NOT", "SHOULD", "SHOULD NOT", "RECOMMENDED", "NOT RECOMMENDED", "MAY", and "OPTIONAL" in this document are to be interpreted as described in [[RFC2119](#)].

The notation used in this memo is the ABNF notation of [[RFC5234](#)] as used by iCalendar [[RFC5545](#)]. Any syntax elements shown below that are not explicitly defined in this specification come from iCalendar [[RFC5545](#)].

3. Modifications to Calendar Components

The following changes to the syntax defined in iCalendar [[RFC5545](#)] are made here. New elements are defined in subsequent sections.

```
calprops /= *(
    ;
    ; The following are OPTIONAL,
    ; but MUST NOT occur more than once.
    ;
    uid / url / timezid /
    refresh / color /
    ;
    ; The following are OPTIONAL,
    ; and MAY occur more than once.
    ;
    name / description / image
    ;
)
```

Daboo

Expires October 9, 2011

[Page 3]

```
eventprop /= *(  
    ;  
    ; The following are OPTIONAL,  
    ; but MUST NOT occur more than once.  
    ;  
    color /  
    ;  
    ; The following are OPTIONAL,  
    ; and MAY occur more than once.  
    ;  
    image / broadcast / conference  
    ;  
    )
```

```
todoprop /= *(  
    ;  
    ; The following are OPTIONAL,  
    ; but MUST NOT occur more than once.  
    ;  
    color /  
    ;  
    ; The following are OPTIONAL,  
    ; and MAY occur more than once.  
    ;  
    image / broadcast / conference  
    ;  
    )
```

```
jourprop /= *(  
    ;  
    ; The following are OPTIONAL,  
    ; but MUST NOT occur more than once.  
    ;  
    color /  
    ;  
    ; The following are OPTIONAL,  
    ; and MAY occur more than once.  
    ;  
    image  
    ;  
    )
```

[4.](#) Properties

Daboo

Expires October 9, 2011

[Page 4]

4.1. NAME Property

Property Name: NAME

Purpose: This property specifies the name of the calendar.

Value Type: TEXT

Property Parameters: IANA, non-standard, alternate text representation, and language property parameters can be specified on this property.

Conformance: This property can be specified multiple times in an iCalendar object. However, each property MUST represent the name of the calendar in a different language.

Description: This property is used to specify a name (a short, one-line description) of the iCalendar object that can be used by calendar user agents when presenting the calendar data to a user. Whilst a calendar only has a single name, multiple language variants can be specified by including this property multiple times with different "LANGUAGE" parameter values on each.

Format Definition: This property is defined by the following notation:

```
name          = "NAME" nameparam ":" text CRLF
nameparam     = *(
    ;
    ; The following are OPTIONAL,
    ; but MUST NOT occur more than once.
    ;
    (";" altrepparam) / (";" languageparam) /
    ;
    ; The following is OPTIONAL,
    ; and MAY occur more than once.
    ;
    (";" other-param)
    ;
    )
```

Example: The following is an example of this property:

NAME:Company Vacation Days

4.2. DESCRIPTION Property

This specification modifies the definition of the "DESCRIPTION" property to allow it to be defined on an iCalendar object. The following additions are made to the definition of this property.

Purpose: This property specifies the description of the calendar.

Conformance: This property can be specified multiple times in an iCalendar object. However, each property **MUST** represent the description of the calendar in a different language.

Description: This property is used to specify a lengthy textual description of the iCalendar object that can be used by calendar user agents when describing the nature of the calendar data to a user. Whilst a calendar only has a single description, multiple language variants can be specified by including this property multiple times with different "LANGUAGE" parameter values on each.

4.3. UID Property

This specification modifies the definition of the "UID" property to allow it to be defined on an iCalendar object. The following additions are made to the definition of this property.

Purpose: This property specifies the persistent, globally unique identifier for the iCalendar object.

Conformance: This property can be specified once in an iCalendar object.

4.4. URL Property

This specification modifies the definition of the "URL" property to allow it to be defined on an iCalendar object. The following additions are made to the definition of this property.

Purpose: This property specifies a URL from where the calendar data was retrieved or where it can be refreshed.

Conformance: This property can be specified once in an iCalendar object.

Description: This property specifies a URL identifying the source of the calendar data and a location from where updates can be retrieved.

4.5. TIMEZONE-ID Property

Property Name: TIMEZONE-ID

Purpose: This property specifies the default time zone identifier for the iCalendar object as a whole.

Value Type: TEXT

Property Parameters: IANA and non-standard property parameters can be specified on this property.

Conformance: This property can be specified once in an iCalendar object.

Description: This property specifies a time zone identifier that represents the default timezone for which floating time or all-day events in the iCalendar object can be assumed to be relative to. It can also be used to choose an initial time zone for use when creating new components in the iCalendar object. A "VTIMEZONE" component having a "TZID" property matching the value specified in this property MUST be present in the iCalendar object.

Format Definition: This property is defined by the following notation:

```
timezid      = "TIMEZONE-ID" timezidparam
               ":" text CRLF
               ;Same value syntax as "TZID" property.
```

```
timezidparam = *(";" other-param)
```

Example: The following is an example of this property:

```
TIMEZONE-ID:America/New_York
```

4.6. REFRESH-INTERVAL Property

Property Name: REFRESH-INTERVAL

Purpose: This property specifies a suggested minimum interval for polling for changes of the calendar data from the original source of that data.

Value Type: DURATION

Property Parameters: IANA and non-standard property parameters can be specified on this property.

Conformance: This property can be specified once in an iCalendar object.

Description: This property specifies a positive duration that gives a suggested minimum polling interval for checking for updates to the calendar data. The value of this property SHOULD be used by calendar user agents to limit the polling interval for calendar data updates to the minimum interval specified.

Format Definition: This property is defined by the following notation:

```
refresh          = "REFRESH-INTERVAL" refreshparam
                  ":" dur-value CRLF
                  ;consisting of a positive duration of time.
```

```
refreshparam = *(";" other-param)
```

Example: The following is an example of this property:

```
REFRESH-INTERVAL:P1W
```

4.7. COLOR Property

Property Name: COLOR

Purpose: This property specifies a color used for displaying the calendar, event, todo, or journal data.

Value Type: INTEGER. The value MUST be three SEMICOLON-separated INTEGER values.

Property Parameters: IANA and non-standard property parameters can be specified on this property.

Conformance: This property can be specified once in an iCalendar object, or "VEVENT", "VTODO", or "VJOURNAL" calendar components.

Description: This property specifies a color that client MAY use when presenting the relevant data to a user. Typically this would appear as the "background" color of events or tasks. The value MUST be an RGB value with integer value components in the range 0..255. If a color is specified on a VEVENT, VTODO or VJOURNAL that SHOULD override any color specified on the enclosing iCalendar object.

Format Definition: This property is defined by the following notation:

```
color           = "COLOR" colorparam ":" colorvalue CRLF
colorparam      = *(";" other-param)
colorvalue      = integer ";" integer ";" integer
                  ; Red, green, and blue values in the range
                  ; 0 - 255.
```

Example: The following is an example of this property:

```
COLOR:255;0;255
```

4.8. IMAGE Property

Property Name: IMAGE

Purpose: This property specifies an image associated with the calendar or a calendar component.

Value Type: The default value type for this property is URI. The value type can also be set to BINARY to indicate inline binary encoded content information. The value MUST refer to or be data with a media type of "image".

Property Parameters: IANA, non-standard, display, inline encoding, and value data type property parameters can be specified on this property. The format type parameter can be specified on this property and is RECOMMENDED for inline binary encoded content information.

Conformance: This property can be specified multiple times in an iCalendar object, or "VEVENT", "VTODO", or "VJOURNAL" calendar components.

Description: This property specifies an image for an iCalendar object or a calendar component via a uri or directly with inline data that can be used by calendar user agents when presenting the calendar data to a user. Multiple properties MAY be used to specify alternative sets of images with, for example, varying media subtypes, resolutions or sizes. When multiple properties are present, calendar user agents SHOULD display only one of them, picking one that provides the most appropriate image quality, or display none. The "DISPLAY" parameter is used to indicate the intended display mode for the image. An "ALTURI" parameter is used to provide a "clickable" image where the URI in the parameter

value can be "launched" by a click on the image in the calendar user agent.

Format Definition: This property is defined by the following notation:

```
image      = "IMAGE" imageparam ( ":" uri ) /  
            (  
              ";" "ENCODING" "=" "BASE64"  
              ";" "VALUE" "=" "BINARY"  
              ":" binary  
            )  
            CRLF  
  
imageparam = *(  
            ;  
            ; The following is OPTIONAL for a URI value,  
            ; RECOMMENDED for a BINARY value,  
            ; and MUST NOT occur more than once.  
            ;  
            (";" fmttypeparam) /  
            ;  
            ; The following are OPTIONAL,  
            ; and MUST NOT occur more than once.  
            ;  
            (";" alturiparam) / (";" displayparam) /  
            ;  
            ; The following is OPTIONAL,  
            ; and MAY occur more than once.  
            ;  
            (";" other-param)  
            ;  
            )
```

Example: The following is an example of this property:

```
IMAGE;DISPLAY=BACKGROUND;FMTPRIME=image/png:htt  
p://example.com/images/party.png
```

4.9. BROADCAST Property

Property Name: BROADCAST

Purpose: This property specifies information for accessing a broadcast of the event or task.

Value Type: URI.

Property Parameters: IANA, non-standard, feature, information, language and region property parameters can be specified on this property.

Conformance: This property can be specified multiple times in a "VEVENT" or "VTODO" calendar component.

Description: This property specifies information for accessing a broadcast system for attendees of a meeting or to-do. This might be a feed: URI [[RFC3496](#)] for a live-blog of the event, or it might be an http: URI [[RFC2616](#)] for a web-based blog, or it might be an rtsp: URI [[RFC2326](#)] for an audio stream.

A broadcast is expected to be a uni-directional communication channel, as opposed to a bi-directional channel which is instead indicated by the "CONFERENCE" property (see [Section 4.10](#)).

The "FEATURE" property parameter is used to describe the key capabilities of the broadcast system to allow a client to choose the ones that give the required level of interaction from a set of multiple properties.

The "INFO" property parameter is used to convey additional details on the use of the URI. For example, access codes for the system.

The "LANGUAGE" property parameter is used to indicate the language associated with the property, and in particular any "INFO" property parameter.

The "REGION" property parameter is used to indicate the region where this property is valid. For example, an audio system could have several streaming servers that appropriate in different geographic regions. This property parameter can be used to differentiate those values so that a location aware device could pick the most appropriate one automatically.

Format Definition: This property is defined by the following notation:


```
broadcast = "BROADCAST" broadparam ":" uri CRLF

broadparam = *(
    ;
    ; The following are OPTIONAL,
    ; and MUST NOT occur more than once.
    ;
    (";" featureparam) / (";" infoparam) /
    (";" languageparam) / (";" regionparam) /
    ;
    ; The following is OPTIONAL,
    ; and MAY occur more than once.
    ;
    (";" other-param)
    ;
    )
```

Example: The following are examples of this property:

```
BROADCAST;FEATURE=AUDIO;REGION=CA;rtsp://stream.ca.example.com/event
BROADCAST;FEATURE=AUDIO;REGION=US;rtsp://stream.us.example.com/event
BROADCAST;FEATURE=AUDIO;REGION=UK;rtsp://stream.uk.example.com/event
BROADCAST;FEATURE=FEED;INFO=Live blog:http://blog.example.com/event
ence.example.com
```

4.10. CONFERENCE Property

Property Name: CONFERENCE

Purpose: This property specifies information for accessing a conferencing system.

Value Type: URI.

Property Parameters: IANA, non-standard, feature, information, language and region property parameters can be specified on this property.

Conformance: This property can be specified multiple times in a "VEVENT" or "VTODO" calendar component.

Description: This property specifies information for accessing a conferencing system for attendees of a meeting or to-do. This might be a tel: URI [[RFC3496](#)] for a telephone-based conference number dial-in (with access codes included), or it might be an http: URI [[RFC2616](#)] for a web-based video chat, or a URI for an instant messaging group chat room. If a specific URI for a conferencing system is not available, a data: URI [[RFC2397](#)]

Daboo

Expires October 9, 2011

[Page 12]

containing a text description can be used.

A conference system is expected to be a bi-directional communication channel, as opposed to a uni-directional "feed" which is instead indicated by the "BROADCAST" property (see [Section 4.9](#)).

The "FEATURE" property parameter is used to describe the key capabilities of the conference system to allow a client to choose the ones that give the required level of interaction from a set of multiple properties.

The "INFO" property parameter is used to convey additional details on the use of the URI. For example, the URIs or access codes for the moderator and attendee of a teleconference system could be different, and the "INFO" property parameter could be used to "tag" each "CONFERENCE" property to indicate which is which.

The "LANGUAGE" property parameter is used to indicate the language associated with the property, and in particular any "INFO" property parameter.

The "REGION" property parameter is used to indicate the region where this property is valid. For example, a telephone based conference system could have several local access numbers that are applicable in different geographic regions. This property parameter can be used to differentiate those values so that a location aware device could pick the most appropriate one automatically.

Format Definition: This property is defined by the following notation:

```
conference = "CONFERENCE" confparam ":" uri CRLF
```

```
confparam = *(  
    ;  
    ; The following are OPTIONAL,  
    ; and MUST NOT occur more than once.  
    ;  
    (";" featureparam) / (";" infoparam) /  
    (";" languageparam) / (";" regionparam) /  
    ;  
    ; The following is OPTIONAL,  
    ; and MAY occur more than once.  
    ;  
    (";" other-param)  
    ;  
    ;
```


)

Example: The following are examples of this property:

```
CONFERENCE;FEATURE=AUDIO;INFO=Moderator dial-in;REGION=US-PA:tel:+
1-412-555-0123,,,654321
CONFERENCE;FEATURE=AUDIO;INFO=Attendee dial-in;REGION=US-PA:tel:+1
-412-555-0123,,,555123
CONFERENCE;FEATURE=AUDIO;INFO=Attendee dial-in;REGION=CA:tel:+1-88
8-555-0456,,,555123
CONFERENCE;FEATURE=CHAT;INFO=Chat room:xmpp:chat-123@confer
ence.example.com
```

5. Property Parameters

5.1. ALTURI Property Parameter

Parameter Name: ALTURI

Purpose: To specify a URI alternative to a property value.

Format Definition: This property parameter is defined by the following notation:

```
alturiparam = "ALTURI" "=" DQUOTE uri DQUOTE
```

Description: This property parameter MAY be specified on "IMAGE" properties.

Example:

```
IMAGE;FMTTYPE=image/png;ALTURI="http://example.com/clicked-image1":http://example.com/images/party.png
```

5.2. DISPLAY Property Parameter

Parameter Name: DISPLAY

Purpose: To specify different ways in which an image for a calendar or component can be displayed.

Format Definition: This property parameter is defined by the following notation:


```
displayparam = "DISPLAY" "="
    ("BADGE" /      ; A small "badge" image
     "BACKGROUND" / ; Use as a background image
     "OVERLAY" /    ; Use as an overlay image
     "BANNER" /     ; Use as a "banner" across the top
     x-name /       ; Experimental type
     iana-token)    ; Other IANA registered type
    ;
    ; Default is BADGE
```

Description: This property parameter MAY be specified on "IMAGE" or "IMAGE" properties. In the absence of this parameter, the value "BADGE" MUST be used for the default behavior. The value determines how a client ought to present an image supplied in iCalendar data to the user.

Values for this parameter are registered with IANA as per [Section 7.3](#). New values can be added to this registry following the procedure outlined in [Section 8.2.1 of \[RFC5545\]](#).

Servers and clients MUST handle x-name and iana-token values they don't recognize by not displaying any image at all.

Example:

```
IMAGE;DISPLAY=BANNER;FMTTYPE=image/png:http://example.com/images/weather-cloudy.png
```

5.3. FEATURE Property Parameter

Parameter Name: FEATURE

Purpose: To specify a feature or features of a conference or broadcast system.

Format Definition: This property parameter is defined by the following notation:

```
featureparam = "FEATURE" "=" featuretext *("," featuretext)
featuretext  = ("AUDIO" /      ; Audio capability
                "CHAT" /      ; Chat or instanst messaging
                "FEED" /      ; Blog or Atom feed
                "SCREEN" /    ; Screen sharing
                "VIDEO" /    ; Video capability
                x-name /     ; Experimental type
                iana-token)  ; Other IANA registered type
```


Description: This property parameter MAY be specified on "BROADCAST" or "CONFERENCE" properties.

Example:

```
BROADCAST;FEATURE=AUDIO:rtsp://audio.example.com/event
CONFERENCE;FEATURE=AUDIO,VIDEO:http://video-chat.example.com/;group-id=1234
```

5.4. INFO Property Parameter

Parameter Name: INFO

Purpose: To specify descriptive text about a property.

Format Definition: This property parameter is defined by the following notation:

```
infoparam = "INFO" "=" paramtext
```

Description: This property parameter MAY be specified on "BROADCAST" and "CONFERENCE" properties.

Example:

```
BROADCAST;INFO=Audio stream;FEATURE=AUDIO:rtsp://
audio.example.com/event
CONFERENCE;INFO="Web video chat, access code=7654
3";:http://video-chat.example.com/;group-id=1234
```

5.5. REGION Property Parameter

Parameter Name: REGION

Purpose: To specify a region or regions where a broadcast or conference system is valid.

Format Definition: This property parameter is defined by the following notation:

```
regionparam = "REGION" "=" regiontext *("," regiontext)
regiontext = paramtext ; an ISO3166-1 or ISO3166-2 code
```

Description: This property parameter MAY be specified on "BROADCAST" or "CONFERENCE" properties.

Example:

```
CONFERENCE;FEATURE=AUDIO;INFO=Moderator dial-in;REGION=US-PA:tel:+1
-412-555-0123,,,654321
```

6. Security Considerations

Several of the new properties or parameters defined by this specification allow reference to "external" URIs. Care MUST be taken when accessing data at external URIs as malicious content could be present. In addition, access to those URIs could be tracked, leading to loss of privacy.

Clients MUST by default prompt users before automatically accessing a broadcast or conference system for the first time to prevent attackers from being able to trigger access to a host computer simply by sending an event invitation.

7. IANA Considerations

7.1. Property Registrations

This document defines the following new iCalendar properties to be added to the registry defined in [Section 8.2.3 of \[RFC5545\]](#):

Property	Status	Reference
NAME	Current	RFCXXXX, Section 4.1
DESCRIPTION	Current	RFC5545 Section 3.8.1.5 , RFCXXXX, Section 4.2
UID	Current	RFC5545 Section 3.8.4.7 , RFCXXXX, Section 4.3
URL	Current	RFC5545 Section 3.8.4.6 , RFCXXXX, Section 4.4
TIMEZONE-ID	Current	RFCXXXX, Section 4.5
REFRESH-INTERVAL	Current	RFCXXXX, Section 4.6
COLOR	Current	RFCXXXX, Section 4.7
IMAGE	Current	RFCXXXX, Section 4.8
BROADCAST	Current	RFCXXXX, Section 4.9
CONFERENCE	Current	RFCXXXX, Section 4.10

7.2. Parameter Registrations

This document defines the following new iCalendar property parameters to be added to the registry defined in [Section 8.2.4 of \[RFC5545\]](#):

Property Parameter	Status	Reference
ALTURI	Current	RFCXXXX, Section 5.1
DISPLAY	Current	RFCXXXX, Section 5.2
FEATURE	Current	RFCXXXX, Section 5.3
INFO	Current	RFCXXXX, Section 5.4
REGION	Current	RFCXXXX, Section 5.5

7.3. Display Types Registry

This document defines the following new iCalendar value registry as per [Section 8.2.6 of \[RFC5545\]](#):

Display Type	Status	Reference
BADGE	Current	RFCXXXX, Section 5.2
BACKGROUND	Current	RFCXXXX, Section 5.2
OVERLAY	Current	RFCXXXX, Section 5.2
BANNER	Current	RFCXXXX, Section 5.2

7.4. Feature Types Registry

This document defines the following new iCalendar value registry as per [Section 8.2.6 of \[RFC5545\]](#):

Feature Type	Status	Reference
AUDIO	Current	RFCXXXX, Section 5.3
CHAT	Current	RFCXXXX, Section 5.3
FEED	Current	RFCXXXX, Section 5.3
SCREEN	Current	RFCXXXX, Section 5.3
VIDEO	Current	RFCXXXX, Section 5.3

8. Acknowledgments

Thanks to the following for feedback: Bernard Desruisseaux, Mike Douglass, Arnaud Quillaud, and Dave Thewlis. This specification came about via discussions at the Calendaring and Scheduling Consortium.

9. References

9.1. Normative References

- [RFC2119] Bradner, S., "Key words for use in RFCs to Indicate Requirement Levels", [BCP 14](#), [RFC 2119](#), March 1997.
- [RFC5234] Crocker, D. and P. Overell, "Augmented BNF for Syntax Specifications: ABNF", STD 68, [RFC 5234](#), January 2008.
- [RFC5545] Desruisseaux, B., "Internet Calendaring and Scheduling Core Object Specification (iCalendar)", [RFC 5545](#), September 2009.

9.2. Informative References

- [RFC2326] Schulzrinne, H., Rao, A., and R. Lanphier, "Real Time Streaming Protocol (RTSP)", [RFC 2326](#), April 1998.
- [RFC2397] Masinter, L., "The "data" URL scheme", [RFC 2397](#), August 1998.
- [RFC2616] Fielding, R., Gettys, J., Mogul, J., Frystyk, H., Masinter, L., Leach, P., and T. Berners-Lee, "Hypertext Transfer Protocol -- HTTP/1.1", [RFC 2616](#), June 1999.
- [RFC3496] Malis, A. and T. Hsiao, "Protocol Extension for Support of Asynchronous Transfer Mode (ATM) Service Class-aware Multiprotocol Label Switching (MPLS) Traffic Engineering", [RFC 3496](#), March 2003.
- [RFC5546] Daboo, C., "iCalendar Transport-Independent Interoperability Protocol (iTIP)", [RFC 5546](#), December 2009.

Appendix A. Change History (To be removed by RFC Editor before publication)

Changes in -04:

1. TZID changed to new property TIMEZONE-ID.
2. Minor formal syntax changes.

Changes in -03:

1. Dropped CALENDAR- prefix
2. DESCRIPTION, UID and TZID now based on existing [RFC5545](#) properties
3. COLOR now on both the calendar and component level
4. IMAGE now on both the calendar and component level
5. Added FEATURE and REGION parameters to CONFERENCE property
6. Added ALTURI parameter to IMAGE property
7. Added FEED value to FEATURE parameter
8. Added BROADCAST property and clarified that CONFERENCE is for bi-direction channels and BROADCAST is for uni-directional.

Changes in -02:

1. Minor wording changes.
2. Interval is now described as the "minimum interval".
3. Added CONFERENCE property and INFO parameter.

Changes in -01:

1. Fixed DISPLAY parameter handling of x- and iana tokens to state that clients ignore the image if the token is not recognized.
2. Allow language variants for CALENDAR-NAME and CALENDAR-DESCRIPTION.
3. Added registry for DISPLAY values.

Author's Address

Cyrus Daboo
Apple Inc.
1 Infinite Loop
Cupertino, CA 95014
USA

Email: cyrus@daboo.name

URI: <http://www.apple.com/>